

**Courbure des tissus plans définis implicitement
par une équation différentielle polynomiale en y' .
Programmation**

J.P. Dufour et D. Lehmann

Abstract :

The aim of this paper is mainly, after some theoretical explanations, to provide a program on Maple for computing, whatever be d , the curvature of the planar d -web implicitly defined by a differential equation $F(x, y, y') = 0$, F being polynomial of degree d with respect to y' .

Moreover, we prove in the appendix a "concentration theorem" for any calibrated ordinary d -web of codimension one in a n -dimensional manifold (in particular for any planar web). Its curvature matrix, relatively to an "adapted" trivialization, is concentrated on the $(n - 2 + k_0)!/(n - 2)!k_0!$ last lines (the last line if $n = 2$), k_0 denoting the integer ≥ 2 such that $d = (n - 1 + k_0!)/(n - 1)!k_0!$.

1 Introduction

A. Hénaut a beaucoup étudié les tissus plans définis par une équation différentielle $F(x, y, y') = 0$, où

$$F(x, y, p) = \sum_{i=0}^d a_i(x, y) p^{d-i}$$

désigne un polynôme de degré d en p , à coefficients a_i dans l'anneau \mathcal{O}_U des fonctions holomorphes sur un ouvert U de \mathbb{C}^2 .

On suppose

- que, pour tout (x, y) dans U , $F(x, y, p)$ et $F'_p(x, y, p)$ ne sont jamais simultanément nuls, c'est-à-dire que les racines $p_i(x, y)$ ($i = 1, \dots, d$) de F , considéré comme polynôme en p , sont toutes distinctes,

- et, pour simplifier les calculs, que le coefficient a_0 de p^d est identiquement égal à 1, soit

$$F(x, y, p) \equiv \prod_{i=1}^d (p - p_i(x, y)).$$

Notons \mathcal{F}_i le feuilletage défini par l'équation différentielle $y' = p_i(x, y)$; en superposant les d feuilletages \mathcal{F}_i , on obtient un d -tissu holomorphe décomposable sur U .

Dans [H] en particulier, A. Hénaut a défini, pour un tel tissu, une certaine connexion dont la courbure généralise à tout d la courbure de Blaschke-Dubourdieu ([B]) du cas $d = 3$, et qui est l'obstruction à ce que le rang du tissu (c'est-à-dire la dimension de l'espace vectoriel de ses relations abéliennes) soit maximum, égal à $\frac{1}{2}(d-1)(d-2)$.

Dans sa thèse, à la même époque, L. Pirio [Pi] a en particulier modernisé des travaux anciens de A. Pantazi [Pa], également en termes de connexion.

Des programmes permettant de calculer cette courbure ont été rédigés sur Maple, pour $d = 3, 4$ ou 5, dans le cas "explicite" par Pirio ([Pi]), et dans le cas "implicite" par O. Ripoll ([R1]).

Dans [CL], nous avons généralisé la définition de cette courbure aux d -tissus de codimension un sur une variété holomorphe de dimension n arbitraire ($n \geq 2$), pourvu que ceux-ci soient "ordinaires" (condition toujours vérifiée pour $n = 2$, et génériquement vérifiée localement pour $n \geq 3$), et "calibrés"

(ce qui veut dire qu'il existe un entier k_0 ($k_0 \geq 2$), tel que $d = \frac{(n-1+k_0)!}{(n-1)!k_0!}$, cette condition étant toujours vérifiée pour $n = 2$ avec $k_0 = d - 1$).

Dans [DL], nous avons rédigé un programme sur Maple, permettant de calculer cette courbure quels que soient n et k_0 , en un temps raisonnable pourvu que n et k_0 ne soient pas "trop grands". Toutefois, pour appliquer ce programme, il fallait disposer d'une intégrale première pour chaque feuilletage \mathcal{F}_i , et par conséquent, lorsque $n = 2$, supposer intégrée chaque équation différentielle $y' = p_i(x, y)$. En fait, nous disposons aussi d'un autre programme (non publié) qui s'applique au cas où chaque feuilletage local du tissu est défini par une forme intégrable non nécessairement fermée, ce qui -pour $n = 2$, revient au cas "explicite" où l'on connaît au moins la décomposition $F(x, y, p) \equiv \prod_{i=1}^d (p - p_i(x, y))$. Dans le cas "implicite" où l'on ne sait pas réaliser cette décomposition, il nous faut généraliser à tout d une méthode de programmation du type de celle proposée par Ripoll dans les cas $d = 3, 4, 5$. C'est l'objet de ce travail¹.

Nous résumons ci-dessous les quelques points de la théorie dont nous avons besoin pour expliquer la programmation, et renvoyons à [Pi], [H], [CL] ou [DL] pour plus de détails.

2 Rappels sur la définition des relations abéliennes

Il revient au même de se donner la différentielle d'une intégrale première du feuilletage \mathcal{F}_i ou une fonction holomorphe $f_i : (x, y) \mapsto f_i(x, y)$ telle que la 1-forme $f_i(dy - p_i dx)$ soit fermée. Une relation abélienne du d -tissu obtenu par superposition des \mathcal{F}_i est alors définie par une famille de fonctions f_i ($i = 1, \dots, d$), telle que

- la somme $\sum_{i=1}^d f_i(dy - p_i dx)$ soit nulle (condition dite "de trace nulle"),
 - chaque forme $f_i(dy - p_i dx)$ est fermée, soit :
- $$(I) \quad \sum_i f_i \equiv 0,$$
- $$(II) \quad \sum_i p_i f_i \equiv 0,$$
- $$(III_i) \quad (f_i)'_x + (p_i f_i)'_y \equiv 0 \text{ pour tout } i.$$

Notons \tilde{U} l'ouvert de la variété de contact au dessus de U , admissible pour les coordonnées (x, y, p) , et S la surface d'équation $F(x, y, p) = 0$ dans \tilde{U} . Soit \mathcal{F} le feuilletage sur S défini par la restriction $(dy - p dx)_S$ à cette surface de la forme de contact $dy - p dx$ sur \tilde{U} : la surface S est un revêtement (trivial) à d feuillets de U . Notant $\pi_i : U_i \xrightarrow{\cong} U$ la projection sur U du feuillet U_i d'équation $p = p_i(x, y)$, la restriction $\tilde{\mathcal{F}}_i$ de $\tilde{\mathcal{F}}$ à U_i se projette sur \mathcal{F}_i par π_i .

Soit $(f_i)_i$ une famille de fonctions sur U . Il revient au même de se donner une fonction f sur S (celle dont la restriction à U_i est égale à $f_i \circ \pi_i$). Et il revient aussi au même de dire que la forme $f(dy - p dx)_S$ sur S est fermée ou que chaque forme $f_i(dy - p_i dx)$ l'est. Mais exprimer cette condition, ainsi que la nullité de la trace pose problème, quand on ne sait pas identifier chaque feuillet $p = p_i(x, y)$.

2.1 La méthode de Hénaut :

Elle consiste à prolonger la fonction $f : S \rightarrow \mathbb{C}$ en une fraction rationnelle sur \tilde{U} de la forme

$$\tilde{f}(x, y, p) = \frac{r(x, y, p)}{F'_p(x, y, p)}$$

(donc sans pôle sur S), r désignant le polynôme d'interpolation en p des expressions $f_i(x, y) \cdot F'_p(x, y, p_i(x, y))$. Son degré est a priori au plus $d - 1$, mais nous allons voir qu'il est même au plus $d - 3$ dès lors que

¹Dans le cas explicite où l'on a le choix, le programme explicite est en général plus performant.

sont vérifiées les relations (I) et (II) :

$$r(x, y, p) = \sum_{j=0}^{d-3} r_j(x, y) p^j.$$

On prend alors comme fonctions inconnues définissant les relations abéliennes les $d - 2$ coefficients r_j de r , ($j = 0, \dots, d - 3$), au lieu des d fonctions f_i ($i = 1, \dots, d$) reliées par les relations (I) et (II).

Notons $\mathcal{O}_k[p]$ le \mathcal{O}_U -module des polynômes $P \in \mathcal{O}_U[p]$, de degré $\leq k$, et soit

$$L_i(x, y, p) = \prod_{j, j \neq i} (p - p_j(x, y))$$

dans $\mathcal{O}_{d-1}[p]$: puisque $F(x, y, p) = \prod_i (p - p_i(x, y))$, F'_p est égal à $\sum_i L_i$. Posons

$$r = \sum_i f_i L_i.$$

Puisque $\frac{r}{F'_p}$ est égal à f_i sur le feuillet U_i , $\frac{r}{F'_p}$ prolonge f à un voisinage de S dans \tilde{U} .

A priori, le degré de r est égal à $d - 1$. Mais le coefficient r_{d-1} est égal à $\sum_i f_i$, tandis que $r_{d-2} = \sum_i p_i f_i + a_1(\sum_i f_i)$. Les relations (I) et (II) deviennent donc $r_{d-1} = 0$, et $r_{d-2} = 0$, c'est-à-dire : $r \in \mathcal{O}_{d-3}[p]$.

Remarque : Au lieu de prolonger la fonction f par la fraction rationnelle $\frac{r}{F'_p}$ ($= \frac{\sum_i f_i L_i}{\sum_i L_i}$), on aurait pu la prolonger par le polynôme d'interpolation $\sum_i \frac{f_i}{L_i(p_i)} L_i = \sum_{j=0}^{d-1} t_j(x, y) p^j$ des f_i , et prendre comme fonctions inconnues les d coefficients t_j reliés par les deux équations (I) et (II) (qui s'expriment facilement à l'aide des seuls coefficients a_i de F). Mais cette méthode est plus compliquée du point de vue de la programmation : si l'on résume les équations (I) et (II) sous la forme

$$(**) \quad < TT(a), t > \equiv 0,$$

($TT(a)$ désignant une matrice $2 \times d$ construite² à partir des coefficients a_i), il faut prendre les dérivées partielles successives de cette identité jusqu'à l'ordre $d - 2$, prendre le noyau du système obtenu, définir une trivialisation de ce noyau, etc... La méthode Hénaut est plus simple, puisque le nombre des fonctions inconnues r_j est déjà réduit à $d - 2$.

Il reste maintenant à exprimer que la restriction à S de la 1-forme $\frac{r(x, y, p)}{F'_p(x, y, p)}(dy - p dx)$ est fermée.

Lemme 1 : *Sur la surface S , est vérifiée l'égalité*

$$d\left(\frac{r(x, y, p)}{F'_p(x, y, p)}(dy - p dx)\right) = \left((r'_x + p r'_y) - \left(\frac{F'_x + p F'_y}{F'_p}\right)'_p \cdot r - \frac{F'_x + p F'_y}{F'_p} r'_p\right) \frac{dx \wedge dy}{F'_p}.$$

Démonstration : Posons $K := F'_x + p F'_y$, d'où $F''_{xp} + p F''_{yp} = K'_p - F'_y$.

Sur \tilde{U} , $d\left(\frac{r(x, y, p)}{F'_p(x, y, p)}(dy - p dx)\right) = d\left(\frac{r(x, y, p)}{F'_p(x, y, p)}\right) \wedge (dy - p dx) + \frac{r(x, y, p)}{F'_p(x, y, p)}(-dp \wedge dx)$, soit :

$$\left[\left(\frac{r(x, y, p)}{F'_p(x, y, p)}\right)'_x + p \left(\frac{r(x, y, p)}{F'_p(x, y, p)}\right)'_y\right] dx \wedge dy + \left(\frac{r(x, y, p)}{F'_p(x, y, p)}\right)'_p dp \wedge (dy - p dx) - \frac{r(x, y, p)}{F'_p(x, y, p)}(dp \wedge dx).$$

²Pour $d = 3$ par exemple, $TT(a) = \begin{pmatrix} (a_1)^2 - 2a_2 & -a_1 & 3 \\ -(a_1)^3 + 3a_1a_2 - 3a_3 & (a_1)^2 - 2a_2 & -a_1 \end{pmatrix}$.

L'expression $\left[\left(\frac{r(x,y,p)}{F'_p(x,y,p)} \right)'_x + p \left(\frac{r(x,y,p)}{F'_p(x,y,p)} \right)'_y \right]$ est égale à $\frac{(r'_x + p r'_y)F'_p - r(K'_p - F'_y)}{(F'_p)^2}$.

D'autre part, $dF \equiv 0$ sur la surface S d'équation $F(x,y,p) = 0$, d'où $dp = \frac{-1}{F'_p}(F'_x dx + F'_y dy)$,
 $dp \wedge dx = \frac{F'_y}{F'_p} dx \wedge dy$ et $dp \wedge dy = -\frac{F'_x}{F'_p} dx \wedge dy$. Ainsi, sur S , $d\left(\frac{r(x,y,p)}{F'_p(x,y,p)}(dy - p dx)\right) = A dx \wedge dy$,

$$\begin{aligned} \text{avec} \quad A &= \frac{(r'_x + p r'_y)F'_p - r.K'_p}{(F'_p)^2} - \left(\frac{r(x,y,p)}{F'_p(x,y,p)} \right)'_p \cdot \frac{K}{F'_p}, \\ &= \frac{1}{(F'_p)^3} \left((r'_x + p r'_y)(F'_p)^2 + r(K.F''_{p^2} - K'_p.F'_p) - r'_p.K.F'_p \right), \\ &= \frac{1}{(F'_p)^3} \left((r'_x + p r'_y)(F'_p)^2 + H.r - L.r'_p \right), \end{aligned}$$

où l'on a posé :

$$H := -(F'_p)^2.(K/F'_p)'_p (= K.F''_{p^2} - K'_p.F'_p), \quad \text{et } L := K.F'_p.$$

Il résulte de ce lemme que la 1-forme $\frac{r(x,y,p)}{F'_p(x,y,p)}(dy - p dx)$ sera fermée sur S ssi :
il existe $c \in \mathcal{O}_{2d-4}[p]$ tel que

$$(*_0) \quad (F'_p)^2.(r'_x + p r'_y) + H.r - L.r'_p \equiv c F.$$

En égalant les coefficients des polynômes de degré $\leq 3d-4$ en p dans $(*_0)$, on obtient un système de $3d-3$ équations à $3d-5$ inconnues qui sont les $d-2$ coefficients r_j de r et les $2d-3$ coefficients c_u de c .

Les $2d-3$ dernières équations³ forment un système cramérien par rapport à ces inconnues parasites que sont les coefficients c_u , ce qui va nous permettre d'éliminer ces derniers en résolvant ce système cramérien et en reportant la solution dans les d premières équations qui ne feront plus intervenir alors que les $d-2$ inconnues r_j ainsi que leurs dérivées. Pour cela, il va être commode d'utiliser des notations matricielles.

2.2 Notations matricielles :

Tout d'abord, aucune confusion n'étant à craindre, on notera souvent de la même façon un polynôme $P = \sum_{i=0}^k g_i(x,y) p^i$ de degré $\leq k$ et le $(k+1)$ -vecteur colonne qui lui correspond, tandis que la multiplication $\mathcal{O}_h[p] \xrightarrow{\times P} \mathcal{O}_{h+k}[p]$ par P sera représentée par la matrice $M(P)$ ci-dessous à $h+k+1$ lignes et $h+1$ colonnes⁴ :

$$P = \begin{pmatrix} g_0 \\ g_1 \\ \vdots \\ g_k \end{pmatrix}, \quad M(P) = \begin{pmatrix} g_0 & 0 & 0 & \cdots & 0 \\ g_1 & g_0 & 0 & \cdots & 0 \\ \vdots & \vdots & \ddots & & \\ g_k & g_{k-1} & & \ddots & 0 \\ 0 & g_k & g_{k-1} & \cdots & g_0 \\ 0 & 0 & g_k & \cdots & g_1 \\ \vdots & \vdots & \vdots & \ddots & \vdots \\ 0 & 0 & 0 & 0 & g_k \end{pmatrix}.$$

³ordonnées dans l'ordre croissant du degré auquel correspondent ces coefficients.

⁴Il n'est pas interdit que g_k soit nul.

Ainsi $r = \sum_{j=0}^{d-3} r_j(x, y)p^j$ sera représenté par un $(d-2)$ -vecteur colonne $\begin{pmatrix} r_0 \\ \vdots \\ r_{d-2} \\ r_{d-3} \end{pmatrix}$, de même que r'_x, r'_y ,
et $r'_p = \langle N, r \rangle$, N désignant la matrice $N = \begin{pmatrix} 0 & 1 & 0 & 0 & \cdots & 0 \\ 0 & 0 & 2 & 0 & \cdots & 0 \\ 0 & 0 & 0 & 3 & \cdots & 0 \\ \vdots & \vdots & \vdots & \vdots & \ddots & \vdots \\ 0 & 0 & 0 & 0 & \cdots & d-3 \\ 0 & 0 & 0 & 0 & \cdots & 0 \end{pmatrix}$, de taille $(d-2) \times (d-2)$.

Considérons en particulier les matrices

$$\begin{aligned} M(F) : \mathcal{O}_{2d-4} [p] &\longrightarrow \mathcal{O}_{3d-4} [p] && \text{de taille } (3d-3) \times (2d-3), \\ M((F'_p)^2) : \mathcal{O}_{d-2} [p] &\longrightarrow \mathcal{O}_{3d-4} [p] && \text{de taille } (3d-3) \times (d-1), \\ M(L) : \mathcal{O}_{d-3} [p] &\longrightarrow \mathcal{O}_{3d-4} [p] && \text{de taille } (3d-3) \times (d-2), \\ \text{et } M(H) : \mathcal{O}_{d-3} [p] &\longrightarrow \mathcal{O}_{3d-4} [p] && \text{de taille } (3d-3) \times (d-2) \\ &&& \text{(la première ligne de } M(H) \text{ étant formée de zéros).} \end{aligned}$$

$$\text{Posons : } I_0 = \begin{pmatrix} 1 & 0 & \cdots & 0 \\ 0 & 1 & \cdots & 0 \\ \vdots & \vdots & \ddots & \vdots \\ 0 & 0 & \cdots & 1 \\ 0 & 0 & \cdots & 0 \end{pmatrix} \text{ et } I^0 = \begin{pmatrix} 0 & 0 & \cdots & 0 \\ 1 & 0 & \cdots & 0 \\ 0 & 1 & \cdots & 0 \\ \vdots & \vdots & \ddots & \vdots \\ 0 & 0 & 0 & 1 \end{pmatrix}, \text{ matrices de taille } (d-1) \times (d-2).$$

La relation $(*_0)$ s'écrit alors :

$$(*_0) \quad \langle M((F'_p)^2), (I_0.r'_x + I^0.r'_y) \rangle + \langle (M(H) - M(L).N), r \rangle = \langle M(F), c \rangle.$$

Cette relation se décompose en

$$(*_0)^+ \quad \langle M^+((F'_p)^2), (I_0.r'_x + I^0.r'_y) \rangle + \langle (M^+(H) - M^+(L).N), r \rangle = \langle M^+(F), c \rangle,$$

et

$$(*_0)^- \quad \langle M^-((F'_p)^2), (I_0.r'_x + I^0.r'_y) \rangle + \langle (M^-(H) - M^-(L).N), r \rangle = \langle M^-(F), c \rangle,$$

M^+ (resp. M^-) désignant les sous-matrices formées avec les d premières lignes (resp. les $2d-3$ dernières) des différentes matrices ci-dessus à $3d-3$ lignes.

La matrice $M^-(F)$, qui est triangulaire avec des 1 sur la diagonale, est inversible. On en déduit :

$$(*_0)^- \quad c = \langle (M^-(F))^{-1}, \langle M^-((F'_p)^2), (I_0.r'_x + I^0.r'_y) \rangle + \langle (M^-(H) - M^-(L).N), r \rangle \rangle,$$

que l'on reporte dans $(*)^+$, de sorte qu'après cette élimination de c , $(*_0)$ devient :

$$\langle B, (I_0.r'_x + I^0.r'_y) \rangle + \langle E, r \rangle = 0, \text{ où}$$

$$B = M^+((F'_p)^2) - M^+(F).(M^-(F))^{-1}.M^-((F'_p)^2),$$

$$E = (M^+(H) - M^+(L).N) - M^+(F).(M^-(F))^{-1}(M^-(H) - M^-(L).N).$$

On vérifie alors le

Lemme 2 : La matrice B , qui s'exprime à l'aide des coefficients a_i et ne fait pas intervenir leurs dérivées, est de rang maximum $d-1$.

Soit T n'importe quelle matrice⁵ inverse à gauche de B (c'est-à-dire telle que $T.B$ soit la matrice identité Id_{d-1}) ; on en déduit le

Théorème 1 : *L'espace des relations abéliennes sur U est isomorphe à l'espace des familles de fonctions $r = (r_j)_j$ ($0 \leq j \leq d-3$), solutions de l'équation⁶ :*

$$(*) \quad < I_0.r'_x + I^0.r'_y > \equiv < M, r > \quad \text{où } M = -T.E.$$

3 Le cas élémentaire $d = 3$:

Dans ce cas, le degré de r est 0 : $r = r(x, y)$ et $r'_p \equiv 0$, tandis que M est une matrice 2×1 , c'est-à-dire un 2-vecteur, dont on notera M_1 et M_2 les composantes. Le système $(*)$ s'écrit alors :

$$r'_x = M_1.r, \quad r'_y = M_2.r.$$

La connexion sur le fibré R_0 de rang 1 des 0-jets de relations abéliennes (dont r est une section) s'écrit $\nabla_x r = r'_x - M_1.r$, $\nabla_y r = r'_y - M_2.r$, et sa courbure est donc

$$k = (M_2)'_x - (M_1)'_y.$$

C'est la courbure de Blaschke-Dubourdieu⁷.

4 Principe de la programmation du calcul de la courbure dans le cas général $d \geq 3$:

De façon générale, on notera $(\dots)'_{\alpha, \beta}$ la dérivée partielle $\frac{\partial^{\alpha+\beta} \dots}{\partial x^\alpha \partial y^\beta}$.

Notons R_k le fibré des relations abéliennes formelles à l'ordre k : R_0 est un fibré de rang $d-2$, dont l'espace des sections $r = (r_j)_{1 \leq j \leq d-3}$ peut-être identifié à $\mathcal{O}_{d-3}[p]$. On réservera la notation $r'_{\alpha, \beta}$ aux jets des véritables relations abéliennes, et l'on notera plutôt

$$u_{\alpha, \beta} = (u_{\alpha, \beta})_j \quad (\alpha + \beta \leq k, \quad 0 \leq j \leq d-3)$$

les éléments constitutifs de R_k .

Les éléments de R_1 au dessus d'un élément $u_{0,0} \in R_0$ sont solutions du système linéaire

$$< I_0, u_{1,0} > + < I^0, u_{0,1} > = < M, u_{0,0} >$$

de rang $d-1$, de $(d-1)$ équations à $2(d-2)$ inconnues $(u_{1,0})_j, (u_{0,1})_j$.

Plus généralement, et pour $k \leq d-3$, les éléments de R_{k+1} au dessus d'un élément $e \in R_k$ sont solutions d'un système linéaire $\Sigma_{k+1}(e)$ de rang $(k+1)(d-1)$ (maximum) : Si $e = (u_{\alpha, \beta})_{\alpha+\beta \leq k}$, le système $\Sigma_{k+1}(e)$ est constitué des

$k+1$ équations $(*)_{\alpha, \beta}$ à valeurs vectorielles (ou $(k+1)(d-1)$ équations à valeurs scalaires), obtenues en dérivant l'identité $(*)$ α fois par rapport à x et β fois par rapport à y , avec $\alpha + \beta = k$,

à $(k+2)(d-2)$ inconnues $(u_{\alpha, \beta})_j, (j = 0, \dots, d-3)$, avec cette fois-ci $\alpha + \beta = k+1$.

⁵Si B_i désigne la sous-matrice carrée de B obtenue en supprimant la i -ème ligne, et si l'indice i_0 a été choisi de façon que B_{i_0} soit inversible, on peut prendre pour T la matrice T_{i_0} construite à partir de $(B_{i_0})^{-1}$ en lui ajoutant une colonne de zéros à la i_0 -ème place. C'est ce qu'on fera dans la programmation qui suit.

⁶Nous essaierons de distinguer, dans la mesure du possible, le signe $' \equiv '$ qui marque une égalité entre fonctions ou germes en un point $m \in U$, et le signe $' = '$ qui marque une égalité entre valeurs ou jets en m .

⁷Voir par exemple [B].

Plus précisément, l'équation $(*_\alpha, \beta)$ s'écrit :

$$< I_0, u_{\alpha+1, \beta} > + < I^0, u_{\alpha, \beta+1} > = \sum_{\gamma=0}^{\alpha} \sum_{\delta=0}^{\beta} \binom{\alpha}{\gamma} \cdot \binom{\beta}{\delta} < M'_{\alpha-\gamma, \beta-\delta}, u_{\gamma, \delta} > .$$

On notera en abrégé $< M, u >_{\alpha, \beta}$ le second membre de cette équation, et l'on posera :

$$M_{\alpha, \beta}^{\gamma, \delta} := \binom{\alpha}{\gamma} \cdot \binom{\beta}{\delta} M'_{\alpha-\gamma, \beta-\delta}.$$

On notera P_{k+1} la partie homogène du système $\Sigma_{k+1}(e)$, constitué par la superposition des expressions $< I_0, u_{\alpha+1, k-\alpha} > + < I^0, u_{\alpha, k+1-\alpha} >$ pour α variant de 0 à $k+1$.

Notation : La plupart des matrices considérées se décomposent en blocs $(d-1) \times (d-2)$ construits

- à partir des matrices I^0, I_0 , pour P_k ,

- et à partir de M et de ses dérivées partielles successives pour les seconds membres des systèmes.

Si (...) désigne une telle matrice à $u(d-1)$ lignes et $v(d-2)$ colonnes, on notera $B_{h, \beta}(\dots)$ le bloc

- des lignes comprises entre $(\beta-1)(d-1)+1$ et $\beta(d-1)$,

- et des colonnes comprises entre $(h-1)(d-2)+1$ et $h(d-2)$.

En particulier, $B_{\beta, \beta}(P_{k+1}) = I_0$, $B_{\beta, \beta+1}(P_{k+1}) = I^0$, $B_{\beta, h}(P_{k+1}) = 0$ si $h \neq \beta, \beta+1$, et l'on démontre aisément le

Lemme 3 : La matrice carrée $P := P_{d-2}$, de taille $(d-1)(d-2) \times (d-1)(d-2)$, est inversible.

Posons $r^{(h)} = (r'_{\alpha, \beta})_{\alpha+\beta=h}$, et $u^{(h)} = (u_{\alpha, \beta})_{\alpha+\beta=h}$.

Le système $\Sigma_{d-2}(e)$, de la forme

$$< P, u^{(d-2)} > = < Q, e > ,$$

est donc cramérien ; sa partie homogène, qui ne contient que des zéros et des 1, ne dépend pas de $e = (u^{(k)})_{0 \leq k \leq d-3}$. Il permet de définir la connexion tautologique de Hénaut sur $\mathcal{E} := R_{d-3}$, dont les sections à dérivée covariante nulle s'identifient aux relations abéliennes.

Soit $s = (u_{\alpha, \beta})_{\alpha+\beta \leq d-3}$ une section de \mathcal{E} : chaque $u_{\alpha, \beta} = (u_{\alpha, \beta})_j$, $(0 \leq j \leq d-3)$ est un $(d-2)$ -vecteur (ou plus exactement une fonction

$$u_{\alpha, \beta} : (x, y) \mapsto u_{\alpha, \beta}(x, y)$$

sur U à valeurs dans les $(d-2)$ -vecteurs).

La connexion tautologique est alors définie par

$$(\nabla_x s)_{\alpha, \beta} = \frac{\partial u_{\alpha, \beta}}{\partial x} - u_{\alpha+1, \beta} \quad \text{si } \alpha + \beta \leq d-4,$$

$$(\nabla_x s)_{\alpha, \beta} = \frac{\partial u_{\alpha, \beta}}{\partial x} - < P^{-1}, < Q, s >>_{\alpha+1, \beta} \quad \text{si } \alpha + \beta = d-3,$$

$$(\nabla_y s)_{\alpha, \beta} = \frac{\partial u_{\alpha, \beta}}{\partial y} - u_{\alpha, \beta+1} \quad \text{si } \alpha + \beta \leq d-4,$$

$$(\nabla_y s)_{\alpha, \beta} = \frac{\partial u_{\alpha, \beta}}{\partial y} - < P^{-1}, < Q, s >>_{\alpha, \beta+1} \quad \text{si } \alpha + \beta = d-3.$$

où $< Q, s >$ désigne le second membre du système $\Sigma_{d-2}(e)$.

Réduction du nombre des inconnues :

A priori, pour définir une section s de \mathcal{E} , il faut se donner $(d-1)(d-2)^2/2$ fonctions $(u_{\alpha,\beta})_j$, avec $0 \leq \alpha + \beta \leq d-3$ et $0 \leq j \leq d-3$. Nous allons montrer que l'on peut en fait réduire ce nombre à $(d-1)(d-2)/2$. Pour cela, nous procéderons en deux étapes :

- dans la première, nous réduirons d'abord ce nombre à $(d-2)^2$ en montrant que, pour un ordre de dérivation $k = \alpha + \beta$ donné, les vecteurs $u_{\alpha,\beta}$ sont déterminés par l'ensemble des vecteurs $(u_{0,h})_{0 \leq h \leq k}$.

- dans la seconde, nous réduirons finalement ce nombre à $(d-1)(d-2)/2$ en montrant que les composantes $(u_{0,h})_j$ de $u_{0,h}$ pour $d-2-h \leq j \leq d-3$ ne dépendent que des vecteurs $(u_{0,k})_{k \leq h-1}$.

Théorème 2 : Une section $s = (u_{\alpha,\beta})_{\alpha+\beta \leq h}$ de R_h est entièrement définie par ses composantes $v^{(k)} := u_{0,k}$ pour $k \leq h$. Plus précisément, $u_{\alpha,\beta}$ s'exprime comme une combinaison linéaire

$$u_{\alpha,\beta} = \sum_{k=0}^{\alpha+\beta} E_{\alpha,\beta}^k v^{(k)},$$

où les matrices $E_{\alpha,\beta}^k$ ne dépendent que de M et de ses dérivées, avec la formule de récurrence (d'abord sur $\alpha + \beta$, puis sur α pour $\alpha + \beta$ fixé) :

$$E_{\alpha+1,\beta}^h = -(J_0 \cdot I^0) \cdot E_{\alpha,\beta+1}^h + \sum_{\gamma=0}^{\alpha} \sum_{\delta=0}^{\beta} (J_0 \cdot M_{\alpha,\beta}^{\gamma,\delta}) \cdot E_{\gamma,\delta}^h,$$

où J_0 désigne la matrice de taille $(d-2) \times (d-1)$ consistant à ajouter une colonne de zéros à droite de la matrice Identité Id_{d-2} , avec la convention $E_{\gamma,\delta}^h = 0$ si $h > \gamma + \delta$.

Démonstration : Puisque $J_0 \cdot I_0 = I_{d-2}$, il suffit d'appliquer la matrice J_0 à chacun des deux membres de l'équation $(*)_{\alpha,\beta}$ ci-dessus, pour obtenir le résultat, .

Corollaire : En particulier, Q se décompose en blocs $B_{\beta,h}(Q)$ de taille $(d-1) \times (d-2)$, avec

$$B_{\beta,h}(Q) = \sum_{\gamma=0}^{d-3-\beta} \sum_{\delta=0}^{\beta} M_{d-3-\beta,\beta}^{\gamma,\delta} \cdot E_{\gamma,\delta}^h, \quad (0 \leq \beta \leq d-2, 0 \leq h \leq d-3).$$

Théorème 3 : Pour tout h , $h \leq d-3$, $v^{(h)} = (v^{(h)})_{j=0 \dots d-3}$ est entièrement défini par l'ensemble $\tilde{v}^{(h)} = (v^{(h)})_{j=0 \dots d-3-h}$ de ses composantes $(u_{0,h})_j$ obtenues pour $0 \leq j \leq d-3-h$, ainsi que par les $\tilde{v}^{(k)}$ pour $k < h$, avec la formule de récurrence

$$v^{(h)} = (-1)^h < K^h, u_{h,0} > + \sum_{k=0}^{h-1} (-1)^k < K^{k-1} \cdot ({}_0J), < M, u >_{k,h-1-k} >,$$

dans laquelle $({}_0J)$ désigne la matrice de taille $(d-2) \times (d-1)$ consistant à ajouter une colonne de zéros à gauche de la matrice Identité I_{d-2} , et $K := ({}_0J) \cdot I_0$.

Démonstration : On observe que $({}_0J) \cdot I^0 = I_{d-2}$, tandis que $< K^n, (w_0, w_1, \dots, w_{d-3}) > = (w_n, w_{n+1}, \dots, w_{d-3}, 0, \dots, 0) >$ pour tout entier $n \leq d-3$, et $K^{d-2} = 0$. Par conséquent, l'expression $(-1)^h < K^h, u_{h,0} >$ de la formule ci-dessus n'intervient pas dans le calcul des composantes $v_j^{(h)}$ pour $j > d-3-h$.

En particulier, appliquant $({}_0J)$ aux deux membres de l'équation initiale $(*)_0$, on obtient

$$v^{(1)} = - < K, u_{1,0} > + < ({}_0J) \cdot M, v^{(0)} >,$$

ce qui prouve le théorème pour $k = 1$.

Plus généralement, appliquant $(-1)^k K^{d-1-h+k} \cdot ({}_0J)$ aux deux membres de l'équation $(*_k, h-1-k)$ et sommant terme à terme les égalités obtenues pour h fixé en faisant varier k de 0 à h , on obtient finalement la formule annoncée.

Par contre, on peut fixer librement les inconnues $(u_{0,h})_j$ pour $0 \leq j \leq d-3-h$. Des théorèmes 2 et 3 résulte alors le

Corollaire :

- (i) Pour $k \leq d-2$, R_k est un espace fibré vectoriel holomorphe de rang $\sum_{j=0}^k (d-2-j)$.
- (ii) La projection $R_{d-2} \rightarrow R_{d-3}$ est un isomorphisme de fibrés vectoriels.
- (iii) En particulier, R_{d-3} est un fibré de rang $(1/2)(d-1)(d-2)$.

Numérotation des inconnues libres, et trivialisation⁸ du fibré R_{d-3} :

Le couple (h, j) tel que $0 \leq h \leq d-3$ et $0 \leq j \leq d-3-h$ va recevoir le numéro d'ordre

$$a = \frac{h(2d-3-h)}{2} + j + 1.$$

On notera alors σ_a (ou $s_j^{(h)}$) la section de R_{d-3} telle que $(u_{0,h})_j = 1$, et $(u_{0,h'})_{j'} = 0$ si $(j', h') \neq (j, h)$, (avec $h' \leq d-3$ et $j' \leq d-3-h'$). D'après le théorème 3 ci-dessus, les sections σ_a , pour a variant de 1 à $(1/2)(d-1)(d-2)$, trivialisent le fibré R_{d-3} .

Ainsi, par exemple, les (j, h) sont rangées dans l'ordre

$$\begin{array}{ll} \sigma_1 = s_0^{(0)} & \text{pour } d = 3, \\ \sigma_1 = s_0^{(0)}, \sigma_2 = s_1^{(0)}, \sigma_3 = s_0^{(1)} & \text{pour } d = 4, \\ \sigma_1 = s_0^{(0)}, \sigma_2 = s_1^{(0)}, \sigma_3 = s_2^{(0)}, \sigma_4 = s_0^{(1)}, \sigma_5 = s_1^{(1)}, \sigma_6 = s_0^{(2)} & \text{pour } d = 5, \\ \text{etc...} & \end{array}$$

Numérotation des lignes de Q , (indexées par (h, j) , $0 \leq h \leq d-3$, $0 \leq j \leq d-3$): :

Le couple (h, j) reçoit le numéro d'ordre

$$a' = h(d-2) + j + 1.$$

(Si $j \leq d-3-h$, $a' = a + \frac{h(h-1)}{2}$).

La forme de connexion $\omega = ((\omega_a^b))$ relative à la trivialisation précédente est donnée par la formule:

$$\omega_a^b = \text{composante de } \nabla \sigma_a \text{ sur } \sigma_b,$$

et la forme de courbure $\Omega = ((\Omega_a^b))$ correspondante s'en déduit par la formule :

$$\Omega_a^b = d\omega_a^b + [\omega, \omega]_a^b.$$

5 Programmation sur Maple 8 :

Merci aux experts en Maple de pardonner aux auteurs leurs maladresses en programmation, et de les leur signaler !

⁸Relativement à cette trivialisation, la seule ligne non nulle de la matrice de courbure est la dernière. Ceci correspond en fait au cas particulier $n = 2$ du théorème 4 démontré en appendice.

```
> restart;
> with(LinearAlgebra);
```

Entrée des données :

(Les données entrées ci-dessous sont celles du 5-tissu de Bol).

Entrée de d :

```
> d:=5;
> interface(rtablesize=(d)*(d)+3);
```

Entrée de F :

```
> apply(a,i,x,y); apply(F,x,y,p);
```

Cas implicite : entrée directe de F par ses coefficients a_i (avec $a_0 = 1$) ;
dans le cas explicite, entrée des p_i : $F = \prod_i (p - p_i)$.

```
> p1 := -1; p2 := 1; p3 := y/(x-1); p4 := y/(x+1); p5 := 2 * x * y/(x^2 + y^2 - 1);
> F:=(p-p1)*(p-p2)*(p-p3)*(p-p4)*(p-p5);
> F:=collect(%,p);
> a(0,x,y):=1;
```

A partir de maintenant, l'écriture du programme ne dépend plus des données introduites (sauf le choix de l'indice $i0$ ci-dessous, permettant de calculer un inverse à gauche de MB)

Première partie : Calcul de la matrice M de l'équation initiale :

$\langle I_0, r'_x \rangle + \langle I^0, r'_y \rangle = \langle M, r \rangle$:

```
> Fp:=diff(F,p);
> Fx:=collect(diff(F,x),p);
> Fy:=collect(diff(F,y),p);
> H:=simplify(diff(Fp,p) * (Fx + p * Fy) - Fp * diff(Fx + p * Fy,p));
> H:=collect(H,p);
> L:=simplify(Fp * (Fx + p * Fy));
> L:=collect(L,p);
> for j to d-2 do for i to j-1 do h(i,j):=0 od od;
> for j to d-2 do for i from j to 2*d-1+j do h(i,j):=simplify(coeff(L,p,i-j)) od od;
> for j to d-2 do for i from 2*d+j to 3*d-3 do h(i,j):=0 od od;
> ML:=Matrix(3*d-3,d-2,h);
> for j to d-2 do for i to j-1 do k(i,j):=0 od od;
> for j to d-2 do for i from j to 2*d-1+j do k(i,j):=simplify(coeff(H,p,i-j)) od od;
> for j to d-2 do for i from 2*d+j to 3*d-3 do k(i,j):=0 od od;
> MH:=Matrix(3*d-3,d-2,k);
> apply(f,i,j); for j to (2*d-3) do for i to j-1 do f(i,j):=0 od od;
> for j to (2*d-3) do for i from j to j+d do f(i,j):=coeff(F,p,i-j) od od;
> for j to (2*d-3) do for i from j+d+1 to 3*d-3 do f(i,j):=0 od od;
> MF:=Matrix(3*d-3,2*d-3,f);
```

```

> Fsup:=DeleteRow(MF,d+1..3*d-3);
> Finf:=DeleteRow(MF,1..d);
> IFinf:=simplify(MatrixInverse(Finf));
> expand((Fp)2);
> collect (
> apply(g2,i,j); for j to (d-1) do for i to j-1 do g2(i,j):=0 od od;
> for j to (d-1) do for i from j to (j+2*d-1) do g2(i,j):=coeff((Fp)2,p,i-j) od od;
> for j to (d-1) do for i from (j+2*d) to 3*d-3 do g2(i,j):=0 od od;
> MF2p:=simplify(Matrix(3*d-3,d-1,g2));
> MB:=simplify(simplify(simplify(DeleteRow(MF2p,d+1..3*d-3))-Fsup.IFinf.simplify(DeleteRow(MF2p,1..d))));

```

Il existe i0 tel que Determinant(DeleteRow(MB,i0)) ne soit pas nul ; (modifier i0 si besoin est, en fonction du tissu introduit au départ).

```

> i0:=d-2;
> Rank(DeleteRow(MB,i0));
> IMB1:=factor(factor(MatrixInverse(DeleteRow(MB,i0))));
> apply(t,i,j); for i to d-1 do t(i,i0):=0 od ;for j to i0-1 do for i to d-1 do t(i,j):=IMB1[i,j] od od;
for j from i0+1 to d do for i to d-1 do t(i,j):=IMB1[i,j-1] od od ;
> T:=simplify(simplify(Matrix(d-1,d,t)));
> apply(n,i,j);for i to d-2 do for j to i do n(i,j):=0 od od; for i to d-2 do n(i,i+1):=i od; for i to d-2
do for j from i+2 to d-2 do n(i,j):=0 od od;
> YY:=Matrix(d-2,d-2,n);
> MHL:=simplify(simplify(MH-ML.YY));
> MC:=simplify(simplify(simplify(DeleteRow((MHL),d+1..3*d-3))-Fsup.IFinf.simplify(DeleteRow((MHL),1..d))));
> M:=-simplify(T.MC);
> apply(MM,i,j);
> for i to d-1 do for j to d-2 do MM(i,j):=simplify(simplify(M[i,j])) od od;

```

Deuxième partie : calcul des prolongements de l'équation initiale :

Calcul de $M'_{a,b}$:

```

> ds:=proc(u,a,b) description "donne la dérivée d'ordre supérieur";
if (a=0 and b=0) then u else
if (evalf(a)>0 and b=0) then simplify(diff(u,x$a)) else
if (a=0 and evalf(b)>0) then simplify(diff(u,y$b)) else
simplify(diff(diff(u,y$b), $a))
fi fi fi end proc;
> dM:=proc (a,b) m(a,b):=(i,j)->ds(MM(i,j),a,b); Matrix(d-1,d-2,m(a,b)) end proc;
> apply(vJo,x,y);
for i to d-2 do for j to i-1 do vJo(i,j):=0 od od;
for i to d-2 do vJo(i,i):=1 od;
for i to d-2 do for j from i+1 to d-1 do vJo(i,j):=0 od od;

```

```

> Jo:=Matrix(d-2,d-1,vJo);
> apply(usup0,i,j);
for j to d-2 do usup0(1,j):=0 od;
for i from 2 to d-1 do for j to i-2 do usup0(i,j):=0 od od;
for i from 2 to d-1 do usup0(i,i-1):=1 od;
for i from 2 to d-1 do for j from i to d-2 do usup0(i,j):=0 od od;
> apply(uinf0,i,j);
for j to d-2 do uinf0(d-1,j):=0 od;
for i to d-2 do
for j to i-1 do
uinf0(i,j):=0 od od;
for i to d-2 do
uinf0(i,i):=1 od ;
for i to d-2 do
for j from i+1 to d-2 do
uinf0(i,j):=0 od od;
> Isup0:=Matrix(d-1,d-2,usup0);
> Iinf0:=Matrix(d-1,d-2,uinf0);

```

Calcul des matrices $E(h,a,b)$ telles que $r'_{a,b} = \sum_{h=0}^{a+b} < E(h,a,b), r'_{0,h} > :$

```

> apply(E,h,a,b);
for h from 0 to d-2 do for n from 0 to h-1 do
E(n,0,h):=Matrix(d-2,d-2,0) od od ;
for a from 0 to d-2 do for b from 0 to d-2 do for h from a+b+1 to d-2 do
E(h,a,b):=Matrix(d-2,d-2,0) od od od ;
for h from 0 to d-2 do E(h,0,h):=IdentityMatrix(d-2) od; for k from 0 to d-3 do for a from 0 to k
do for h from 0 to k+1 do
E(h,a+1,k-a):= simplify(-Jo.Isup0.E(h,a,k-a+1)+
sum('binomial(a,c)*sum('binomial(k-a,e)*Jo.dM(a-c,k-a-e).E(h,c,e)', 'e'=0..k-a)', 'c'=0..a)): od od
od;
> for k from 0 to d-3 do for a from 0 to k do for h from 0 to k+1 do
print('E'(h,a+1,k-a)=E(h,a+1,k-a)) od od od;

```

Calcul des matrices $G(h,a,b)$ telles que $< M, r >_{a,b}' = \sum_{h=0}^{a+b} < G(h,a,b), r'_{0,h} > :$

```

> apply(G,h,a,b);
> for k from 0 to d-3 do for a from 0 to k do for h from 0 to k do
G(h,a,k-a):= simplify(simplify(
sum('binomial(a,c)*sum('binomial(k-a,e)*dM(a-c,k-a-e).E(h,c,e)', 'e'=0..k-a)', 'c'=0..a))); od od od;
> for k from 0 to d-3 do for a from 0 to k do for h from 0 to k do
print('G'(h,a,k-a)=G(h,a,k-a)) od od od;

```

Expression du système $\langle P, (r'_{a,b})_{a+b=d-2} \rangle = \langle Q, (r'_{a,b})_{a+b < d-2} \rangle :$

Calcul de la matrice inversible P , partie homogène du système, et matrice inverse $IP :$

```

> apply(u,i,j);
> for a from 1 to d-2 do for i from (a-1)*(d-1)+1 to a*(d-1) do for b from 1 to a-1 do for j from
(b-1)*(d-2)+1 to b*(d-2) do
  u(i,j):=0 ; od od od od;
> for a from 1 to d-2 do for i from (a-1)*(d-1)+1 to a*(d-1) do for j from (a-1)*(d-2)+1 to a*(d-2)
do
  u(i,j):=uinf0(i-(a-1)*(d-1),j-(a-1)*(d-2)) od od od;
> for a from 1 to d-2 do for i from (a-1)*(d-1)+1 to a*(d-1) do for j from a*(d-2)+1 to (a+1)*(d-2)
do
  u(i,j):=usup0(i-(a-1)*(d-1),j-a*(d-2)) od od od;
> for a from 1 to d-2 do for i from (a-1)*(d-1)+1 to a*(d-1) do for b from a+2 to d-1 do for j from
(b-1)*(d-2)+1 to b*(d-2) do
  u(i,j):=0 od od od od;
> P:=Matrix((d-1)*(d-2),(d-2)*(d-1),u);
> IP:=MatrixInverse(P);

```

Calcul de $Q :$

```

> apply(q,i,j);
> for b from 0 to d-3 do for h from 0 to d-3 do for j from h*(d-2)+1 to (h+1)*(d-2) do for i from
b*(d-1)+1 to (b+1)*(d-1) do
  q(i,j):=simplify(G(h,d-3-b,b))[i-b*(d-1),j-h*(d-2)] od od od od ;
> Q:=Matrix((d-2)*(d-1),(d-2)^2,q);
> U:=IP.Q;

```

Expression des inconnues liées $v(h,j)$ pour j variant de $d-2-h$ à $d-3$, en fonction des inconnues libres $v(h,j)$ pour j variant de 0 à $d-3-h :$

```

> apply(voJ,x,y);
> for i to d-2 do for j to i do voJ(i,j):=0 od od;
> for i to d-2 do voJ(i,i+1):=1 od;
> for i to d-2 do for j from i+2 to d-1 do voJ(i,j):=0 od od;
> oJ:=Matrix(d-2,d-1,voJ);
> J:=(oJ).Iinf0;

```

Numérotation des inconnues libres : $(h,j) \rightarrow a = h * (2 * d - 3 - h) / 2 + j + 1, (0 \leq j \leq d - 3 - h) :$

```

> apply(hh,a);
> for a to (d-1)*(d-2)/2 do for h from 0 to d-3 do
  if h*(2*d-3-h)/2 < a and a <= (h+1)*(2*d-4-h)/2 then hh(a):=h fi od od;
> for a to (d-1)*(d-2)/2 do print('hh'(a)=hh(a)) od;
> apply(jj,a); for a to (d-1)*(d-2)/2 do jj(a):=a-1-hh(a)*(2*d-3-hh(a))/2 od;

```

Numérotation de toutes les inconnues : $(h, j) \rightarrow a = h * (d - 2) + j + 1, (0 \leq j \leq d - 3) :$

```
> apply(hhh,a);
> for a to (d-2)*(d-1) do for h from 0 to d-2 do
if h*(d-2)<a and a<=(h+1)*(d-2) then hhh(a):=h fi od od;
> for a to (d-2)*(d-1) do print('hhh'(a)=hhh(a)) od;
> apply(jjj,a); for a to (d-2)*(d-1) do jjj(a):=a-1-hhh(a)*(d-2) od;
```

Trivialisation du fibré porteur de la connexion par les sections $s(a) = (s(a, h, j))_{h,j}, (j \leq d - 3 - h) :$

```
> apply(s,a,h,j);
> for a to (d-1)*(d-2)/2 do for h from 0 to d-3 do for j from 0 to d-3-h do
s(a,hh(a),jj(a)):=1 od od od ;
> for a to (d-1)*(d-2)/2 do for h from 0 to d-3 do for j from 0 to d-3-h do
if h<>hh(a) or j<>jj(a) then s(a,h,j):=0 fi od od od ;
```

Expression des inconnues liées ($j > d - 3 - h$) en fonction des inconnues libres (j au plus $d - 3 - h$) :

```
> for a to (d-1)*(d-2)/2 do for h from 1 to d-3 do for j from d-2-h to d-3 do
s(a,h,j):= sum(' sum(' sum('
(-1)^k*(J^k.oJ.G(n,k,h-1-k))[j+1,r]*s(a,n,r-1)', 'r'=1..d-2)', 'k'=0..h-1)', 'n'=0..h-1)
od od od ;
> for a to (d-1)*(d-2)/2 do for h from 0 to d-3 do for j from 0 to d-3 do print('s'(a,h,j)=s(a,h,j))
od od od;
> apply(WW,a,h);
> for a to (d-1)*(d-2)/2 do for h from 0 to d-3 do WW(a,h):=Vector(d-2) od od;
for a to (d-1)*(d-2)/2 do for h from 0 to d-3 do for i to d-2 do
WW(a,h)[i]:=s(a,h,i-1) od od od;
> for a to (d-1)*(d-2)/2 do for h from 0 to d-3 do
print('WW'(a,h)=WW(a,h)) od od;
> apply(WWW,a);
> for a to (d-1)*(d-2)/2 do WWW(a):=Vector((d-2)^2) od;
> for a to (d-1)*(d-2)/2 do for i to (d-2)^2 do
WWW(a)[i]:=WW(a,hhh(i))[i-hhh(i)*(d-2)] od od;
> for a to (d-1)*(d-2)/2 do
print('WWW'(a)=WWW(a)) od;
```

Troisième partie : expression de la connexion et calcul de sa courbure :

(Relativement à la trivialisation précédente, la forme de connexion est écrite $(Ax) dx + (Ay) dy$)

Dérivation covariante en x des sections $s(a)$; calcul de Ax :

```
> apply(Nablax,a,h,j);
> for a to (d-1)*(d-2)/2 do for h from 0 to d-4 do for j from 0 to d-3 do
Nablax(a,h,j):=simplify(diff(s(a,h,j),x)-sum(' (E(k,1,h).WW(a,k))[j+1]', 'k'=0..h+1)) od od od;
> for a to (d-1)*(d-2)/2 do for h from 0 to d-4 do for j from 0 to d-3 do
```

```

print('Nablax'(a,h,j)=Nablax(a,h,j)) od od od;
> for a to (d-1)*(d-2)/2 do for j from 0 to d-3 do
Nablax(a,d-3,j):=diff(s(a,d-3,j),x)- (U.WWW(a))[(d-3)*(d-2)+j+1] od od;
> for a to (d-1)*(d-2)/2 do for j from 0 to d-3 do
print('Nablax'(a,d-3,j)=Nablax(a,d-3,j)) od od;
> apply(NNx,b,a);
> for b to (d-1)*(d-2)/2 do for a to (d-1)*(d-2)/2 do
NNx(b,a):=simplify(Nablax(a,hh(b),jj(b))) od od;
> Ax:=Matrix((d-1)*(d-2)/2,(d-1)*(d-2)/2,NNx);

```

Dérivation covariante en y ; calcul de Ay :

```

> apply(Nablay,a,h,j);
> for a to (d-1)*(d-2)/2 do for h from 0 to d-4 do for j from 0 to d-3 do
Nablay(a,h,j):=simplify(diff(s(a,h,j),y)-s(a,h+1,j)) od od od;
> for a to (d-1)*(d-2)/2 do for h from 0 to d-4 do for j from 0 to d-3 do
print('Nablay'(a,h,j)=Nablay(a,h,j)) od od od;
> for a to (d-1)*(d-2)/2 do for j from 0 to d-3 do
Nablay(a,d-3,j):=simplify(diff(s(a,d-3,j),y) - (U.WWW(a))[(d-2)2 + j + 1]) od od;
> for a to (d-1)*(d-2)/2 do for j from 0 to d-3 do
print('Nablay'(a,d-3,j)=Nablay(a,d-3,j)) od od;
> apply(NNy,b,a);
> for b to (d-1)*(d-2)/2 do for a to (d-1)*(d-2)/2 do
NNy(b,a):=Nablay(a,hh(b),jj(b)) od od;
> Ay:=Matrix((d-1)*(d-2)/2,(d-1)*(d-2)/2,NNy);

```

Calcul de la courbure :

```

> apply(axy,b,a);
for a to (d-1)*(d-2)/2 do for b to (d-1)*(d-2)/2 do
axy(b,a):=simplify(diff(Nablay(a,hh(b),jj(b)),x)) od od;
> apply(ayx,b,a); for a to (d-1)*(d-2)/2 do for b to (d-1)*(d-2)/2 do
ayx(b,a):=simplify(diff(Nablax(a,hh(b),jj(b)),y)) od od;
> Axy:=simplify(Matrix((d-1)*(d-2)/2,(d-1)*(d-2)/2,axy));
> Ayx:=simplify(Matrix((d-1)*(d-2)/2,(d-1)*(d-2)/2,ayx));
> KK:=simplify(simplify(simplify(Axy)-simplify(Ayx))+
simplify(simplify(Ax).simplify(Ay))-simplify(simplify(Ay).simplify(Ax)));

```

Dans le cas d'une déformation du tissu l'aide du paramètre z (sinon KO=KK) :

```

> ko:=(i,j)->taylor(KK[i,j],z,1);
> KO:=Matrix((d-1)*(d-2)/2,(d-1)*(d-2)/2,ko);

```

6 Appendice : Concentration de la matrice de courbure relative à une trivialisation adaptée

Nous nous placerons plus généralement dans le contexte d'un d -tissu calibré ordinaire⁹ de codimension un dans une variété holomorphe de dimension n ($n \geq 2$).

Soit k_0 l'entier ≥ 2 tel que $d = (n - 1 + k_0)!/(n - 1)!(k_0)!$ (si $n = 2$, $k_0 = d - 1$).

On définit alors une filtration décroissante de $\mathcal{E} = R_{k_0-2}$ en posant :

$$F_0(\mathcal{E}) = \mathcal{E} \text{ et } F_h(\mathcal{E}) = \text{Ker}(R_{k_0-2} \rightarrow R_{h-1}) \text{ pour } 1 \leq h \leq k_0 - 1,$$

chaque $F_h(\mathcal{E})$ étant alors un sous-fibré vectoriel de rang $\sum_{k=h+1}^{k_0-1} (d - c(n, h))$, où l'on a posé

$$c(n, h) := (n - 1 + h)!/(n - 1)!h!.$$

Une trivialisation holomorphe (σ_a) de \mathcal{E} $\left(1 \leq a \leq \sum_{h=1}^{k_0-1} (d - c(n, h))\right)$ sera dite *adaptée* si, pour tout h ($1 \leq h \leq k_0 - 2$), les $(d - c(n, h + 1))$ sections σ_a telles que

$$\sum_{k=1}^h (d - c(n, k)) < a \leq \sum_{k=1}^{h+1} (d - c(n, k))$$

engendrent un supplémentaire S_h (nécessairement holomorphe) de $F_{h+1}(\mathcal{E})$ dans $F_h(\mathcal{E})$.

Théorème 4 : *Relativement à une trivialisation "adaptée", la matrice de courbure de la connexion tautologique d'un d -tissu holomorphe ordinaire calibré de codimension un dans une variété holomorphe de dimension n ($n \geq 2$) est concentrée¹⁰ dans les $(n - 2 + k_0)!/(n - 2)!k_0!$ dernières lignes (la dernière ligne si $n = 2$), où k_0 désigne l'entier ≥ 2 tel que $d = (n - 1 + k_0)!/(n - 1)!k_0!$.*

Cela veut dire que les lignes qui précèdent n'ont que des 0.

Démonstration : Soit $L = (\alpha_1, \dots, \alpha_n)$ un multi-indice de dérivation, et $L + 1_i$ le multi-indice obtenu en augmentant α_i de 1. Supposons la trivialisation (σ_a) adaptée. Une section s de \mathcal{E} est alors représentée par une famille $(s^{(h)})_h$ d'éléments $s^{(h)} \in S_h$, $(s^{(h)}) = (s_L)_{|L|=h}$.

De même, soit $\hat{s}^{(k_0-1)} = (\hat{s}_L)_{|L|=k_0-1}$ la famille des dérivées partielles d'ordre $k_0 - 1$ des composantes de la projection sur R_0 d'une section \hat{s} de R_{k_0-1} (composantes relatives à la trivialisation locale de R_0 déjà utilisée pour les dérivées d'ordre $\leq k_0 - 2$).

D'autre part, le tissu étant ordinaire et calibré, la projection $P : R_{k_0-1} \xrightarrow{\cong} R_{k_0-2}(=\mathcal{E})$ est un isomorphisme de fibrés vectoriels. Le système linéaire exprimant qu'une section \hat{s} de R_{k_0-1} est un $(k_0 - 1)$ -jet de relation abélienne se projetant sur une section s de $R_{k_0-2}(=\mathcal{E})$ s'écrit :

$$< P, \hat{s}^{(k_0-1)} > = < Q, s >$$

pour un certain opérateur linéaire Q . La dérivation covariante de la connexion tautologique sur \mathcal{E} est donc définie par les formules :

$$(\nabla_i s)_L = \partial_i(s_L) - s_{L+1_i} \text{ pour } |L| \leq k_0 - 3,$$

$$\text{et } (\nabla_i s)_L = \partial_i(s_L) - < (P^{-1} \cdot Q), s >_{L+1_i} \text{ pour } |L| = k_0 - 2.$$

On en déduit :

$$(\nabla_i \nabla_j s)_L = \partial_i \partial_j(s_L) - \partial_i(s_{L+1_j}) - \partial_j(s_{L+1_i}) + s_{L+1_i+1_j} \text{ pour } |L| \leq k_0 - 4,$$

⁹On se réfère à [DL] pour la terminologie et les notations.

¹⁰Ce rsultat probablement déjà bien connu pour $n = 2$.

et $(\nabla_i \nabla_j s)_L = \partial_i \partial_j (s_L) - \partial_i (s_{L+1_j}) - \partial_j (s_{L+1_i}) + < (P^{-1}.Q), s >_{L+1_i+1_j}$ pour $|L| = k_0 - 3$.

Dans les deux cas, l'expression est symétrique en i et j : la matrice de courbure ne peut donc avoir de composante non nulle que pour $|L| = k_0 - 2$. Puisque S_{k_0-2} est de rang $c(n, k_0) - c(n, k_0 - 1)$, soit $c(n - 1, k_0)$, le théorème est démontré.

Pour $n = 2$, $c(n - 1, k_0)$ est égal à 1 quel que soit $k_0 = d - 1$.

References

- [B] W. Blaschke, *Über die Tangenten einer ebenen Kurve fünfter Klasse*. Abh. Math. Semin. Hamb. Univ. 9 (1933) 313-317.
- [CL] V. Cavalier, D. Lehmann, *Ordinary holomorphic webs of codimension one*. arXiv 0703596v2 [mathsDS], 2007, et Ann. Sc. Norm. Super. Pisa, cl. Sci (5), vol XI (2012), 197-214. .
- [DL] J. P. Dufour, D. Lehmann, *Calcul explicite de la courbure des tissus calibrés ordinaires* arXiv 1408.3909v1 [mathsDG], 18/08/2014.
- [H] A. Hénaut, *Planar web geometry through abelian relations and connections* Annals of Math. 159 (2004) 425-445.
- [H1] A. Hénaut, *Introduction to planar web geometry*, prépublication d'extraits d'une monographie en cours de rédaction, décembre 2014.
- [Pi] L. Pirio, *Equations Fonctionnelles Abéliennes et Géométrie des tissus*, Thèse de doctorat de l'Université Paris VI, 2004.
- [Pa] A. Pantazi. *Sur la détermination du rang d'un tissu plan*. C.R. Acad. Sc. Roumanie 4 (1940), 108-111.
- [R] O. Ripoll, *Géométrie des tissus du plan et équations différentielles* Thèse de doctorat de l'Université de Bordeaux 1, 2005.
- [R1] O. Ripoll, *Programmation sur Maple pour d=3, 4, 5*, communiqué par A. Hénaut, non publié, (2007).

Jean-Paul Dufour, ancien professeur à l'Université de Montpellier II,
1 rue du Portalet, 34820 Teyran, France
email : dufourh@netcourrier.com,

Daniel Lehmann, ancien professeur à l'Université de Montpellier II,
4 rue Becagrun, 30980 Saint Dionisy, France
email : lehm.dan@gmail.com,